
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of RRC test case 8.1.3.8 which are part of the LTE test suite.

The test case can be demonstrated to run with 1 LTE UE (see section 5). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_8_1_3_8
Test Group:
RRC

ATS Version:
iwd-EUTRA-TVB2009-12_D11wk04
System Simulator used:
Anite Conformance Toolset
UE(s) used:                       ICERA Espresso 300
Verification Status:
PASS
4 Corrections required for test case 8.1.3.8

4.1 Introduction

This section describes the changes required to make test case 8.1.3.8 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-12_D10wk04 release.

4.2 Change 1

Testcase name
cas_CreateGPRSCh

Reason for change
During GPRS channel creation (f_GERAN_CreatePhyGPRSCh()) the template cas_GERAN_CreateGPRSCh is used. The current definition of this template sets the field channelcombination to tsc_PDTCH_f (=18). This is incorrect as this constant refers to a logical channel rather than a channel combination.

Summary of change
Replace tsc_PDTCH_f (=18) with PDTCH (=13) to indicate the correct channel combination.

Source of change
GERAN_Templates.ttcn

Before:

  // =============================================================================

  // ASP Constraint

  // PDTCH/F  + PACCH/F + PTCHH/F

  // -----------------------------------------------------------------------------

  template (value) G_CPHY_CONFIG_REQ cas_CreateGPRSCh(GERAN_CellId_Type p_CellId,

                                                      PhysicalChId p_PhyChId,

                                                      template (value) FreqInfo p_FreqInfo,

                                                      template (value) TN p_TimeSlot,

                                                      template (value) TSC p_TSC,

                                                      TX_Power p_DownlinkPowerLevel,

                                                      B1_Type p_BandIndicator) :=

  {

    createPhyCh := {

      cellId          := enum2int(p_CellId),

      physicalChId    := p_PhyChId,                    // identifier of the physical channel created

      channelCombination := tsc_PDTCH_F,  // logical channels combined onto the basic physical channel

      freqInfo        := p_FreqInfo,

      timeSlot        := p_TimeSlot,                   // the time slot number of the physical channel

      tSC             := p_TSC,

      channelSpecificPara := omit,

      txPower         := p_DownlinkPowerLevel,

      bandIndicator   := p_BandIndicator

    }

  };

After:

  // =============================================================================

  // ASP Constraint

  // PDTCH/F  + PACCH/F + PTCHH/F

  // -----------------------------------------------------------------------------

  template (value) G_CPHY_CONFIG_REQ cas_CreateGPRSCh(GERAN_CellId_Type p_CellId,

                                                      PhysicalChId p_PhyChId,

                                                      template (value) FreqInfo p_FreqInfo,

                                                      template (value) TN p_TimeSlot,

                                                      template (value) TSC p_TSC,

                                                      TX_Power p_DownlinkPowerLevel,

                                                      B1_Type p_BandIndicator) :=

  {

    createPhyCh := {

      cellId          := enum2int(p_CellId),

      physicalChId    := p_PhyChId,                    // identifier of the physical channel created

      channelCombination := enum2int(PDTCH),  // logical channels combined onto the basic physical channel

      freqInfo        := p_FreqInfo,

      timeSlot        := p_TimeSlot,                   // the time slot number of the physical channel

      tSC             := p_TSC,

      channelSpecificPara := omit,

      txPower         := p_DownlinkPowerLevel,

      bandIndicator   := p_BandIndicator

    }

  };

4.3 Change 2

Testcase name
cs_G_CommonCellConfigInfoDef

Reason for change
The template cs_G_CommonCellConfigInfoDef omits value for IE ‘gamma’ which is later passed as a parameter to template cs_PktUplinkAss_IARO, used in the transmission of Immediate Assignment message.

In template cs_PktUplinkAss_IARO, since ‘allocMask’ is present the presence of ‘gamma’ is mandatory.

.

Summary of change
A value ‘00100’B is assigned to ‘gamma’ as is in current UTRAN implementation

Source of change
GERAN_CellInfo.ttcn

Before:

template (value) GERAN_CommonCellInfo_Type cs_G_CommonCellConfigInfoDef (O2_Type  p_CellIdentity):=

  {

    gamma := omit,

    downlinkPowerLevel := 63,                        // Downlink transmission power level Vemf().

    cellIdentity    := p_CellIdentity,               // cell identity

    mcc             := '001'H,                       // mobile country code = 001 (decimal)

    mnc             := '01F'H,                       // mobile network code = 01 (decimal)

    plmn            := '00F110'O,                    // PLMN

    lac             := '0001'O,                      // location area code = '0001'O

    rac             := '05'O,

…

}

After:

template (value) GERAN_CommonCellInfo_Type cs_G_CommonCellConfigInfoDef (O2_Type  p_CellIdentity):=

  {

    gamma := '00100'B,

    downlinkPowerLevel := 63,                        // Downlink transmission power level Vemf().

    cellIdentity    := p_CellIdentity,               // cell identity

    mcc             := '001'H,                       // mobile country code = 001 (decimal)

    mnc             := '01F'H,                       // mobile network code = 01 (decimal)

    plmn            := '00F110'O,                    // PLMN

    lac             := '0001'O,                      // location area code = '0001'O

    rac             := '05'O,

…

}

4.4 Change 3

Testcase name
f_GERAN_GPRS_Authentication

Reason for change
In function f_GERAN_GPRS_Authentication(),  the response parameter, v_AuthRES is checked against XRES string generated at SS from bit zero.

However, it has to compared from bit 32 as it has the RES Ext stored from the Authentication and Ciphering Response message received from UE.

.

Summary of change
Compare strings from bit 32

Source of change
GERAN_CommonFunctions.ttcn

Before:

function f_GERAN_GPRS_Authentication (GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

…

// Verify that the received Authentication Response parameters match expected response:

          if (v_AuthRES != substr(v_SecurityInfo.authPS.XRES, 0, lengthof(v_AuthRES))) {

            f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");

            }

…

}

After:

function f_GERAN_GPRS_Authentication (GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

…

// Verify that the received Authentication Response parameters match expected response:

          if (v_AuthRES != substr(v_SecurityInfo.authPS.XRES, 32, lengthof(v_AuthRES))) {

            f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");

            }

…

}

4.5 Change 4

Testcase name
f_GERAN_GPRSAttach

Reason for change
During GPRS attach procedure controlled by function f_GERAN_GPRSAttach(), the TTCN does not handle the indication of release of the uplink TBF by the UE, following the transmission & reception of ATTACH COMPLETE. This message can result in a FAIL or INCONCLUSIVE verdict if not handled.

Summary of change
Add code to the end of this function (duplicated from the beginning of f_DownlinkTBFEstablishment()) to handle the release of the final uplink TBF by the UE.

Source of change
GERAN_CommonFunctions.ttcn

Before:

function f_GERAN_GPRSAttach(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_AttachType;

    var octetstring v_PLMN := f_GERAN_PLMN_Get(p_CellId);

    var O1_Type v_RAC := f_GERAN_RAC_Get(p_CellId);

    var O2_Type v_LAC := f_GERAN_LAC_Get(p_CellId);

………

//Receive GPRS Attach Complete Message

    f_UplinkTBFOnePhase (p_CellId);

    G_LLC.receive(car_G_LLC_UnitData_IND (cr_G_AttachComplete));

}// end of GPRSAttach

After:

function f_GERAN_GPRSAttach(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_AttachType;

    var octetstring v_PLMN := f_GERAN_PLMN_Get(p_CellId);

    var O1_Type v_RAC := f_GERAN_RAC_Get(p_CellId);

    var O2_Type v_LAC := f_GERAN_LAC_Get(p_CellId);

    timer t_Dly := 0.250;
…………..

//Receive GPRS Attach Complete Message

    f_UplinkTBFOnePhase (p_CellId);

    G_LLC.receive(car_G_LLC_UnitData_IND (cr_G_AttachComplete));

    //Anite - handle TBF release indication

    t_Dly.start;

    alt {

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement))

        {

          t_Dly.stop;

        }

      [] t_Dly.timeout {}

    }
  }// end of GPRSAttach

4.6 Change 5

Testcase name
f_GERAN_RAU

Reason for change
During GPRS Routing Area Update procedure controlled by function f_GERAN_RAU(), the TTCN does not handle the indication of release of the uplink TBF by the UE, following the transmission & reception of ROUTING AREA UPDATE COMPLETE. This message can result in a FAIL or INCONCLUSIVE verdict if not handled.

Summary of change
Add code to the end of this function (duplicated from the beginning of f_DownlinkTBFEstablishment()) to handle the release of the final uplink TBF by the UE.

Source of change
GERAN_CommonFunctions.ttcn

Before:

function f_GERAN_RAU(GERAN_CellId_Type p_CellId,

                       B1_Type p_FOR,

                       B3_Type p_Type) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    var octetstring v_PLMN := f_GERAN_PLMN_Get(p_CellId);

    var O1_Type v_RAC := f_GERAN_RAC_Get(p_CellId);

    var O2_Type v_LAC := f_GERAN_LAC_Get(p_CellId);

……

//Receive RAU Complete Message

    f_UplinkTBFOnePhase (p_CellId);

    G_LLC.receive(car_G_LLC_UnitData_IND (cr_G_RA_UpdComplete));

  }// end of f_GERAN_RAU

}

After:

function f_GERAN_GPRSAttach(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_AttachType;

    var octetstring v_PLMN := f_GERAN_PLMN_Get(p_CellId);

    var O1_Type v_RAC := f_GERAN_RAC_Get(p_CellId);

    var O2_Type v_LAC := f_GERAN_LAC_Get(p_CellId);

    timer t_Dly := 0.250;
…………..

//Receive RAU Complete Message

    f_UplinkTBFOnePhase (p_CellId);

    G_LLC.receive(car_G_LLC_UnitData_IND (cr_G_RA_UpdComplete));

    //Anite - handle TBF release indication

    t_Dly.start;

    alt {

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement))

        {

          t_Dly.stop;

        }

      [] t_Dly.timeout {}

    }
  }// end of f_GERAN_RAU

4.7 Change 6

Testcase name
fl_ConvertGERANNeighCell

Reason for change
In the function fl_ConvertGERANNeighCell(), where the code loops through a list of GERAN ARFCNs, the loop condition is incorrect.

Summary of change
Change “<=” to “<” in the loop condition

Source of change
GERAN_CommonFunctions.ttcn

Before:

function fl_ConvertGERANNeighCell (CoOrd_GERANNeighCell_Type p_NeighCell_Type) return ExplicitListOfARFCNs

  {

    var ExplicitListOfARFCNs v_ListOfARFCNs ;

    var integer i;

    for (i:=0; i <= lengthof(p_NeighCell_Type); i:= i+1) {

      if ( (p_NeighCell_Type[i]>=0) and (p_NeighCell_Type[i] <= 1023)) {    // Check the value is in range of ARFCN-ValueGERAN ::=  INTEGER (0..1023)

        v_ListOfARFCNs[i] := p_NeighCell_Type[i];

      }

    };

    return v_ListOfARFCNs;

  };

After:

function fl_ConvertGERANNeighCell (CoOrd_GERANNeighCell_Type p_NeighCell_Type) return ExplicitListOfARFCNs

  {

    var ExplicitListOfARFCNs v_ListOfARFCNs ;

    var integer i;

    for (i:=0; i < lengthof(p_NeighCell_Type); i:= i+1) {

      if ( (p_NeighCell_Type[i]>=0) and (p_NeighCell_Type[i] <= 1023)) {    // Check the value is in range of ARFCN-ValueGERAN ::=  INTEGER (0..1023)

        v_ListOfARFCNs[i] := p_NeighCell_Type[i];

      }

    };

    return v_ListOfARFCNs;

  };

4.8 Change 7

Testcase name
f_GERAN_508CheckCampOnNewGeranCell

Reason for change
When performing RAU on a new GERAN cell, the TTCN should accept an update type of “RA updating” or “Combined RA / LA updating” to allow UEs operating in either PS-only or CS / PS modes to pass the test case.

Summary of change
Allow update type of 0 or 1 in RAU request. Test suite constants are used in line with the function f_UTRAN_GMM_RAU()

Source of change
GERAN_CommonFunctions.ttcn

Before:

function f_GERAN_508CheckCampOnNewGeranCell ( GERAN_CellId_Type p_CellId ) runs on GERAN_PTC

  {

    //  @desc Procedure which aims at checking whether the UE is camping on a new GERAN cell

    //  of a test case or not. The procedure is defined in 36.508 table 6.4.2.9-1.

    f_GERAN_RAU (p_CellId, '0'B, '001'B);
    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "");

  }

After:

function f_GERAN_508CheckCampOnNewGeranCell ( GERAN_CellId_Type p_CellId ) runs on GERAN_PTC

  {

    //  @desc Procedure which aims at checking whether the UE is camping on a new GERAN cell

    //  of a test case or not. The procedure is defined in 36.508 table 6.4.2.9-1.

    f_GERAN_RAU (p_CellId, '0'B, (tsc_I_AttachTypeGPRSAtt, tsc_I_AttachTypeCombined)); 

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "");

  }

4.9 Change 8

Testcase name
cs_GPRS_CellOptionsSI13

Reason for change
In the GPRS Cell Options IE in SI 13, the “extension length” field is set to 3 bits which does not allow the extension fields specified in cs_GPRS_CellOptionsExtension to be encoded

Summary of change
Set extension length to 10 bits

Source of change
GERAN_templates.ttcn

Before:

// =============================================================================

  // Structured Type Constraint

  // values from TS 51.010-1 v.5.5.0 subclause 40.2.1.1.1

  // -----------------------------------------------------------------------------

  template (value) GPRS_CellOptions cs_GPRS_CellOptionsSI13(template (value) B2_Type p_NMO) :=

  {

    nMO             := p_NMO,                        // network operation  mode I

    t3168           := '011'B,                       // 2 second

    t3192           := '010'B,                       // 1.5 second

    dRX_TIMER_MAX   := '000'B,                       // non-DRX not supported, 0 second

    aCCESS_BURST_TYPE := '1'B,                       // 11 bits

    cONTROL_ACK_TYPE := '1'B,                        // RLC/MAC control block

    bS_CV_MAX       := '0111'B,                      // 7

    pAN_Mask        := '1'B,

    pAN_DEC         := '011'B,                       // 3

    pAN_INC         := '011'B,                       // 3

    pAN_MAX         := '010'B,                       // N3102=12

    optionalExtMask := '1'B,                         // for R99

    extensionLength := ‘000011’B,

    extensionInfo   := cs_GPRS_CellOptionsExtension

  };



After:

// =============================================================================

  // Structured Type Constraint

  // values from TS 51.010-1 v.5.5.0 subclause 40.2.1.1.1

  // -----------------------------------------------------------------------------

  template (value) GPRS_CellOptions cs_GPRS_CellOptionsSI13(template (value) B2_Type p_NMO) :=

  {

    nMO             := p_NMO,                        // network operation  mode I

    t3168           := '011'B,                       // 2 second

    t3192           := '010'B,                       // 1.5 second

    dRX_TIMER_MAX   := '000'B,                       // non-DRX not supported, 0 second

    aCCESS_BURST_TYPE := '1'B,                       // 11 bits

    cONTROL_ACK_TYPE := '1'B,                        // RLC/MAC control block

    bS_CV_MAX       := '0111'B,                      // 7

    pAN_Mask        := '1'B,

    pAN_DEC         := '011'B,                       // 3

    pAN_INC         := '011'B,                       // 3

    pAN_MAX         := '010'B,                       // N3102=12

    optionalExtMask := '1'B,                         // for R99

    extensionLength := '001010'B,       //10
    extensionInfo   := cs_GPRS_CellOptionsExtension

  };

4.10 Change 9

Testcase name
cs_GPRS_CellOptionsExtension

Reason for change
In the extension bits of the GPRS Cell Options IE in SI 13, the R7 extension “Reduced latency access” is included, but previous R4 and R6 extensions are set to OMIT. These fields need to be included for the IE to be encoded correctly.

Summary of change
Include all R4 & R6 extensions in GPRS Cell Options IE

Source of change
GERAN_templates.ttcn

Before:

  template (value) ExtensionInfo cs_GPRS_CellOptionsExtension :=

  {

    extensionMask   := '0'B,                         // EGPRS not supported

    eGPRS_PACKET_CHANNEL_REQUEST := omit,            // EGPRS not supported

    bEP_PERIOD      := omit,                         // EGPRS not supported

    pFC_FEATURE_MODE := '0'B,                        // packet context procedure not suppoprted

    dTM_SUPPORT     := '0'B,                         // DTM not supported

    bSS_PAGING_COORDINATION := '0'B,                 // CS paging coordination not supported

    cCN_ACTIVE      := omit,                         // R4 extension

    nW_EXT_UTBF     := omit,

    multiTBFCap     := omit,                         // R6 extension

    extUTBFNoData   := omit,

    dTMEhancCap     := omit,

    mbmsMask        := omit,
    dedModeMBMSSupprt := omit,                       // not present if mbmsMask = 0

    mNCISupport     := omit,                         // not present if mbmsMask = 0

    reducedLatencyAccess := '0'B,

    spareBit        := omit

  };



After:

  template (value) ExtensionInfo cs_GPRS_CellOptionsExtension :=

  {

    extensionMask   := '0'B,                         // EGPRS not supported

    eGPRS_PACKET_CHANNEL_REQUEST := omit,            // EGPRS not supported

    bEP_PERIOD      := omit,                         // EGPRS not supported

    pFC_FEATURE_MODE := '0'B,                        // packet context procedure not suppoprted

    dTM_SUPPORT     := '0'B,                         // DTM not supported

    bSS_PAGING_COORDINATION := '0'B,                 // CS paging coordination not supported

    cCN_ACTIVE      := '0'B,                         // R4 extension

    nW_EXT_UTBF     := '0'B,

    multiTBFCap     := '0'B,                         // R6 extension

    extUTBFNoData   := '0'B,

    dTMEhancCap     := '0'B,

    mbmsMask        := '0'B,
    dedModeMBMSSupprt := omit,                       // not present if mbmsMask = 0

    mNCISupport     := omit,                         // not present if mbmsMask = 0

    reducedLatencyAccess := '0'B,

    spareBit        := omit

  };

4.11 Change 10

Testcase name
8.1.3.8

Reason for change
GERAN PTC should send interRAT sys info to the E-UTRAN PTC before waiting to receive authentication parameters so that EUTRAN PTC can execute preamble. This is required as before starting the preamble on EUTRA, EUTRA PTC expect sys info to update SIB 7 as per the specific message content required for this TC.

Summary of change
Adjust order of statements

Source of change
RRC_ConnRel_GERAN.ttcn

Before:

  //RRC Connection Release: redirection from E-UTRAN to GERAN

  function f_TC_8_1_3_8_GERAN () runs on GERAN_PTC

  {

    var boolean v_TriggerRxd := false;

    var integer v_T0_RSSI_Cell24 := -70;

    //Initialise test case

    f_GERAN_Init ( EUTRA_GERAN );

    //Create GERAN cell and broadcast System Infomation

    f_GERAN_CreateCell ( geran_Cell24 );

    f_GERAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    //* @desc Coordinate with E-UTRA component to send GERAN frequencies

    f_GERAN_Send_IR_SysInfoToEUTRA ( geran_Cell24 );
    //Initial trigger: wait for trigger to set power level according to T0 and finish preamble

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA);

    ……………

  }; //end of testcase 8.1.3.8



After:

//RRC Connection Release: redirection from E-UTRAN to GERAN

  function f_TC_8_1_3_8_GERAN () runs on GERAN_PTC

  {

    var boolean v_TriggerRxd := false;

    var integer v_T0_RSSI_Cell24 := -70;

    //Initialise test case

    f_GERAN_Init ( EUTRA_GERAN );

    //Create GERAN cell and broadcast System Infomation

    f_GERAN_CreateCell ( geran_Cell24 );

    //* @desc Coordinate with E-UTRA component to send GERAN frequencies

    f_GERAN_Send_IR_SysInfoToEUTRA ( geran_Cell24 );
    f_GERAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    //Initial trigger: wait for trigger to set power level according to T0 and finish preamble

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA);

    ……………

  }; //end of testcase 8.1.3.8

4.12 Change 11

Testcase name
8.1.3.8

Reason for change
EUTRAN PTC is sending iRAT coordination trigger too early, and authentication parameters are not being sent at all to GERAN PTC. 

Summary of change
Send authentication parameters to GERAN PTC at the end of preamble, followed by IRAT coordination trigger

Source of change
RRC_ConnRel.ttcn

Before:

  function f_TC_8_1_3_8_EUTRA ( ) runs on EUTRA_PTC {

………..

    //Initial trigger: tell GERAN side to finish preamble

    f_SendIRATCoOrd ( GERAN, cms_IRAT_Trigger );
    //Create and configure all cells

    f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    //Bring UE to initial state

    f_EUTRA_Preamble ( eutra_Cell1, STATE2_IDLEUPDATE );

    f_EUTRA_RbEst_Def ( eutra_Cell1 );

    //Set cell power level according to T0

    f_EUTRA_SetCellPower ( eutra_Cell1, v_T0_Cell_Specific_RS_EPRE_Cell1, cs_TimingInfo_Now );

…….

}



After:

function f_TC_8_1_3_8_EUTRA ( ) runs on EUTRA_PTC {

………..

    //Create and configure all cells

    f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    //Bring UE to initial state

    f_EUTRA_Preamble ( eutra_Cell1, STATE2_IDLEUPDATE );

    f_EUTRA_RbEst_Def ( eutra_Cell1 );

    //Set cell power level according to T0

    f_EUTRA_SetCellPower ( eutra_Cell1, v_T0_Cell_Specific_RS_EPRE_Cell1, cs_TimingInfo_Now );

    // then get latest security params and send them over

    f_EUTRA_SendAuthParameters (eutra_Cell1, GERAN);   

    //Initial trigger: tell GERAN side to finish preamble

    f_SendIRATCoOrd ( GERAN, cms_IRAT_Trigger );
…….

}

5 Execution Log Files.

5.1 ICERA Espresso E300

The ICERA Espresso E300 UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_8_1_3_8_ICERA-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

[1]
R5s110082: This archive comprises html format execution log file.

